AD NUMBER

AD911536

NEW LIMITATION CHANGE

TO

Approved for public release, distribution unlimited

FROM

Distribution authorized to U.S. Gov't. agencies only; Test and Evaluation; MAY 1973. Other requests shall be referred to Department of the Air Force, Directorate of Operational Requirements and Development Plans, Attn: RDQSD, Washington, DC 20330.

AUTHORITY

USAF ltr, 14 Jan 1974

AD911536

DEFENSE DIVISION NOTE

DDN 73-2

QUICK EVALUATION COMPUTER MODELS FOR AIR DEFENSE

May 1973

Victor V. Gogolak

DISTRIBUTION LIMITED,
to U.S. Government agencies only; test and evaluation; May 1973. Other requests for this document must be referred to Hq USAF (AF/RDQSD).
Wash, D.C. 20330

ANALYTIC SERVICES INC. (ANSER)

This Division Note reports research sponsored by the *Directorate of Operational Requirements and Development Plans, DCS/Research and Development, Hq USAF, under Contract F44620-69-C-0014.* It does not necessarily represent the opinion of anyone other than the author and the Division Manager. It may be withdrawn at any time.

DEFENSE DIVISION NOTE

DDN 73-2

QUICK EVALUATION COMPUTER MODELS FOR AIR DEFENSE

May 1973

Victor V. Gogolak

Approved by

R. S. Timm, Manager

Wash, D.C. 20330

Analytic Services Inc. (ANSER)
5613 Leesburg Pike, Falls Church, Virginia 22041

ABSTRACT

This Note discusses three time-sharing computer programs that are quick tools for air defense engagement analysis. The programs are based on a Monte Carlo method. The model MONTYX evaluates air defense interceptor configurations by determining their effectiveness against a bomber cell in the terminal phase of an intercept. The model DLMNTY expands the capability of MONTYX to a situation where two bomber types are contained within the bomber cell. The results are bomber probabilities of survival for various interceptor-to-bomber ratios. The model TOOTH involves penetration by four types of bombers—two carrying gravity weapons and two carrying standoff weapons. The defense is provided by two types of interceptors in two waves. In the first wave, only raid count by cell is provided. The second wave assumes that a bomber count has been made and standoff weapons have been launched. The results indicate the numbers of bombers of each type surviving each wave of defense as well as the number of standoff weapons launched and those which survive the second wave of defense.

TABLE OF CONTENTS

I. INTRODUCTION

Models developed for air defense engagement analysis range from the exponential interceptor/bomber (I/B) formulas to massive and expensive simulations of the strategic air war. Some of these examine enemy penetrations of U.S. air defenses and others examine U.S. bomber penetration.

The models presented here have been in use for over five years and have not grown in complexity or sophistication. That is not their purpose. They are quick tools which fill the gap between the formulas and the war game simulations. As analytic tools, they all require a certain amount of homework to establish the inputs. In this way, some fairly detailed analyses can be made, with the programs taking the burden of the calculations.

The programs are written in XBASIC and can be run on any time-sharing computer system having an XBASIC compiler. All programs are based on a Monte Carlo method. This allows programs to handle such problems as imperfect commitment with only a few decision statements. Closed form solutions are not used.

Applications have varied from parametric analysis of the probability of detection and conversion (P_{DC}) and probability of kill (P_K) for interceptors to assessment of equivalent megatonage (EMT) delivered on targets in massive exchanges. A fair amount of "art" is involved in setting up the inputs to allow the analyst close control of the games.

II. SINGLE WAVE BOMBER INTERCEPTOR MODEL—MONTYX

A. Purpose

The purpose of the model called MONTYX is to evaluate air defense interceptor configurations (aircraft, missiles, and fire control systems) by determining their effectiveness against a bomber cell in the terminal phase of an intercept.

B. Assumptions

The bombers and interceptors involved in the interaction are assumed to have the same configuration; i.e., the bombers represent identical targets and the interceptors have the same values for probabilities of detection and conversion and of kill. The simulation begins after initial detection has taken place; the interceptors are assigned to bombers on a random basis. The group of interceptors may make any specified number of passes (P) against the bomber cell and are assumed to carry at least P weapons. An interceptor may fire one weapon per pass but retains its weapon if it has been assigned to a bomber already killed. The model is run for various interceptor/bomber ratios. The game may be played many times to improve statistics. The number of times the game is played is specified in the input and the results are averaged.

C. Program Description

1. Input

When the program is run, the following numbers are requested from the user:

- Number of passes (P)
- Probabilities of detection and conversion and of kill for each pass: "1" P(DC), P(K); "2" P(DC), P(K).
- Interceptor-to-bomber ratios—low, high, and step size I1, I2, I9, where I1 is the lowest I/B ratio and I2 is the highest ratio and I9 is the increment which steps the game from I1 to I2. For example, (I1, I2, I9) = (1, 2, .5) would indicate that the game would be run for I/B = 1, 1.5, 2.
- Loop size (M1)—which indicates the number of times the game is to be played; the final results are the averages of the M1 games.

2. Simulation

The simulation begins after the inputs are specified by the user. The model assigns a number to each bomber. On each pass, each interceptor is randomly assigned to a bomber; that is, an interceptor randomly chooses a bomber to attack. This bomber has not been killed on a previous pass but could have been killed earlier in the same pass by another interceptor. The bomber is considered killed or not killed by a random choice based on the probabilities of detection and conversion and of kill involved. At the end, the total number of bombers surviving is tallied. After the specified number of passes, some interceptors (those which did not convert on a target during one or more of the passes) will still have weapons. These are left unused. The model provides a number of runs based on the interceptor-to-bomber ratios (I/B) and prints out the bomber probability of survival (P_S) corresponding to the I/B ratio.

3. Output

The output consists of the initial conditions of the game (specified by the user) and the bomber (P_S) for the various I/B ratios.

4. Coding

- Program List for MONTYX
- Instructions for MONTYX
- Variable List for MONTYX

PROGRAM LIST FOR MONTYX

```
100 DIM F(50),B(10)
110 PRINT "NUMBER ØF PASSES?";
120 INPUT P
130 FØR P1=1 TØ P
140 PRINT "PASS";P1;"P(DC),P(K)?";
150 INPUT D(P1),L(P1)
160 NEXT P1
170 PRINT "I/B RANGE (LØW,HIGH,STEP)?";
180 INPUT I1,I2,I9
182 PRINT "LØØP SIZE?";
184 INPUT M1
190 FØR I=10*I1 TØ 10*I2 STEP 10*I9
200 S=0
230 FØR J=1 TØ M1
240 FØR K=1 TØ 10
250 B(K)=1
260 NEXT K
270 B1=10
280 FØR K=1 TØ P
290 FØR K1=1 TØ I
300 F(K1)=0
310 NEXT K1
320 FØR K1=1 TØ I
330 IF RND(0)>D(K) THEN 370
340 K2=INT(10*RND(0)+1)
350 IF B(K2)=0 THEN 340
360 F(K1)=K2
370 NEXT K1
380 FØR K1=1 TØ I
390 IF F(K1)=0 THEN 460
400 LET X=F(K1)
410 IF B(X)=0 THEN 460
420 IF RND(0)>L(K) THEN 460
430 B(X)=0
440 B1=B1-1
450 IF B1=0 THEN 510
460 NEXT K1
470 NEXT K
490 S=S+B1
510 NEXT J
520 PRINT I/10,S/(10*M1)
530 NEXT I
540 PRINT
550 PRINT "ANØTHER GAME?";
560 INPUT G$
570 IF G$="YES" THEN 110
1000 END
```

INSTRUCTIONS FOR MONTYX

100 Dimensions the bomber and interceptor arrays.

110-184 Input section; values of variables are requested and entered from the terminal.

- 110-120 Number of passes.
- 130-160 Loop indexed by number of passes for entering probabilities of detection and conversion and of kill for each pass.
- 170-180 I/B range.
- 182-184 Loop size for iteration of games for each I/B.

198-530 Outer loop indexed for I/B.

- 200 Sum of surviving bombers is set to 0.
- 230-510 Iteration loop for better statistics.
  - 240-260 Bomber array is set to 1 (all bombers alive).
  - 270 Total bombers in a game initialized to 10.
  - 280-470 PASS LOOP.
    - 290-310 Interceptors initialized to 0.
    - 320-370 Interceptor assignment loop.
      - 330 Check if detection and conversion occur.
      - 340 Pick a random number, 1-10, and assign an interceptor to that bomber.
      - 350 If bombe. is dead, pick another number.
      - 360 Set interceptor array to bomber number.
    - 370 NEXT ASSIGNMENT.
    - 380-460 ATTACK LOOP. Index by interceptor.
      - 390 If interceptor unassigned, next interceptor
      - 410 If bomber already killed, next interceptor.
      - 420 Check if kill occurs. (If not, next interceptor.)

430 Set bomber to 0.

440 Decrease total bombers by 1.

450 If all bombers in that game are dead, begin next iteration.

460 NEXT ATTACK

470 NEXT PASS

490 Add surviving bombers to total survivors.

510 NEXT ITERATION

520 Prints out I/B and probability of survival.

530 NEXT I/B

550-570 Check for another game.

580 END

VARIABLE LIST FOR MONTYX

F(*)	Interceptor Array
B(*)	Bomber Array
P1	Loop Index
K	Loop Index
K1	Loop Index
I	Loop Index
J	Loop Index
I1	Lowest I/B
I2	Highest I/B
I9	Step size for I/B
M1	Loop size
P	Number of passes
D(*)	Probability of Detection and Conversion Array
L(*)	Probability of Kill Array
S	Sum of surviving bombers within an I/B run

5. Sample Runs of MONTYX

```
NUMBER ØF PASSES?? 1
PASS 1     P(DC),P(K)?? .5,.5
I/B RANGE (LØW,HIGH,STEP)?? .2,2,.2
LØØP SIZE?? 100
 .2                   .98
 .4                   .913
 .6                   .878
 .8                   .837
 1.                   .788
 1.2                  .737
 1.4                  .728
 1.6                  .684
 1.8                  .629
 2.                   .611

ANØTHER GAME?? YES
NUMBER ØF PASSES?? 1
PASS 1     P(DC),P(K)?? .7,.5
I/B RANGE (LØW,HIGH,STEP)?? .2,2,.2
LØØP SIZE?? 100
 .2                   .968
 .4                   .879
 .6                   .815
 .8                   .786
 1.                   .729
 1.2                  .664
 1.4                  .614
 1.6                  .61
 1.8                  .543
 2.                   .506

ANØTHER GAME?? YES
NUMBER ØF PASSES?? 1
PASS 1     P(DC),P(K)?? .5,.7
I/B RANGE (LØW,HIGH,STEP)?? .2,2,.2
LØØP SIZE?? 100
 .2                   .965
 .4                   .908
 .6                   .835
 .8                   .782
 1.                   .724
 1.2                  .681
 1.4                  .623
 1.6                  .575
 1.8                  .568
 2.                   .532

ANØTHER GAME?? YES
```

```
NUMBER ØF PASSES?? 3
PASS 1     P(DC),P(K)?? .7,.7
PASS 2     P(DC),P(K)?? .6,.6
PASS 3     P(DC),P(K)?? .5,.5
I/B RANGE (LØW,HIGH,STEP)?? .2,2,.2
LØØP SIZE?? 100
 .2                    .894
 .4                    .679
 .6                    .497
 .8                    .348
 1.                    .202
 1.2                   .127
 1.4                   5.50000E-02
 1.6                   2.20000E-02
 1.8                   1.00000E-02
 2.                    4.00000E-03

ANØTHER GAME?? NØ
```

6. Plot Subroutine—MONTYP

MONTYP consists of a group of statements that when merged with MONTYX, plot rather than tabulate the results. The basic program, the logical basis, and the input procedure remain the same.

To run the plot version, MONTYX and MONTYP are merged into a program MONTYZ, which is then run.

```
190 PRINT
191 PRINT TAB(2);"PS";TAB(40);"PROBABILITY OF SURV_VAL"
192 PRINT TAB(17);"0";TAB(27);".1";TAB(37);".2";TAB(47);".3";TAB(57);".[illegible]
193 PRINT TAB(67);".5";TAB(77);".6";TAB(87);".7";TSB(097);".8";
194 PRINT TAB(107);".9";TAB(117);"I"
195 PRINT TAB(16);"I----+----I----+----I----+----I----+----I----+";
196 PRINT "----I----+----I----+----I----+----I----+----I----+----I"
515 PRINT TAB(16);"I"
520 PRINT TAB(16);"I"
528 PRINT TAB(16);"I"
530 PRINT,1000,S/(10*M1),I/10;
532 PRINT" +";
534 PRINT TAB(17+INT(100*((S/(10*M1))+.005)));"*"
540 NLXT I
542 PRINT
544 PRINT
546 PRINT
1000 FMT F5.3,F10.2
9999 END
```

SAMPLE RUNS USING MONTYP
(Merge MONTYZ, MONTYX, MONTYP)



SAMPLE RUNS USING MONTYP—Continued
(Merge MONTYZ, MONTYX, MONTYP)



SAMPLE RUNS USING MONTYP—Continued
(Merge MONTYZ, MONTYX, MONTYP)

```
                YES
NUMBER OF PASSES?!1
PASS      1 P(DC),P(K)?!.5,.7
I/B RANGE (LOW,HIGH,STEP)?!.2,2,.2
LOOP SIZE?!100
```




```
ANOTHER GAME?!
```

SAMPLE RUNS USING MONTYP—Continued
(Merge MONTYZ, MONTYX, MONTYP)

```
              YES
NUMBER OF PASSES?!3
PASS      1 P(OC),P(K)?!.7,.7
PASS      2 P(OC),P(K)?!.6,.6
PASS      3 P(OC),P(K)?!.5,.5
I/B RANGE (LOW,HIGH,STEP)?!.2,2,.2
LOOP SIZE?!100
```




```
ANOTHER GAME?!NO
 9999 EXIT
?
```

III. SINGLE-WAVE DUAL BOMBER-INTERCEPTOR MODEL—DLMNTY

A. Purpose

The purpose of DLMNTY is to expand the capability of MONTYX to consider a situation wherein two bomber types are contained within the bomber cell.

B. Program Description

1. Simulation-Input-Output

The assumptions and operation of DLMNTY are identical to those of MONTYX; the only differences are in the input and output.

The following numbers are requested as input from the user:

- P = number of passes.
- W = bomber scaling factor. For example, W = 2 indicates there are twice as many bombers of type A as of type B.
- Loop size M, which indicates the number of times the game is to be played.
- Probabilities of detection and conversion and of kill for each pass against each bomber type.
- Interceptor-to-bomber ratio—low, high, and step size. I7, I9, and I8, where I7 is the lowest I/B ratio, I9 is the highest ratio, and I8 is the increment which steps the game from I7 to I9.

The output consists of the initial conditions of the games (specified by the user) and the probabilities of survival (P_S) for the total bomber cell as well as the P_S for each bomber type.

2. Coding

- Program List for DLMNTY
- Instructions for DLMNTY
- Variable List for DLMNTY

PROGRAM LIST FOR DLMNTY

```
100 DIM F(200),B(50)
110 DIM A(50)
120 PRINT "WHAT ARE PASSES, BØMBER SCALE FACTØR, AND LØØP SIZE?";
130 INPUT P,W,M1
140 IF P=0 THEN 1010
150 PRINT "NEW PRØBABILITIES ARRAY?(YES-1,NØ-0)";
160 INPUT Y
170 IF Y=0 THEN 220
180 FØR P1=1 TØ P
190 PRINT "PASS";P1;"P(DC),P(K) (BØMBER A),P(DC),P(K) (BØMBER B)";
200 INPUT C(P1),M(P1),D(P1),L(P1)
210 NEXT P1
220 PRINT "I/B RANGE (LØW,HIGH,AND STEP)?";
230 INPUT I7,I9,I8
240 PRINT TAB(10);"I/B";TAB(25);"PRØBABILITY ØF SURVIVAL"
250 PRINT TAB(29);"ALL BØMBERS      TYPE A         TYPE B"
310 FØR I=I7*10*(W+1) TØ I9*10*(W+1) STEP I8*10*(W+1)
320 S=0
330 T=0
340 FØR J=1 TØ M1
350 FØR K=1 TØ 10
360 B(K)=1
370 NEXT K
380 FØR K=1 TØ 10*W
390 A(K)=1
400 NEXT K
410 B1=10
420 A1=W*10
430 FØR K=1 TØ P
440 FØR K1=1 TØ I
450 F(K1)=0
460 NEXT K1
470 FØR K1=1 TØ I
480 IF RND(0)>1/(1+W)THEN 545
485 IF B1=0 THEN 600
490 IF RND(0)>D(K) THEN 600
510 K2=INT(10*RND(0)+1)
520 IF B(K2)=0 THEN 510
530 F(K1)=K2
540 GØ TØ 600
```

PROGRAM LIST—Continued

```
545 IF A1=0 THEN 600
550 IF RND(0)>C(K) THEN 600
570 K3=INT(W*10*RND(0)+1)
580 IF A(K3)=0 THEN 570
590 F(K1)=100+K3
600 NEXT K1
610 FØR K1=1 TØ I
620 IF F(K1)=0 THEN 770
630 IF F(K1)>100 THEN 700
640 LET X=F(K1)
650 IF B(X)=0 THEN 770
660 IF RND(0)>L(K) THEN 770
670 B(X)=0
680 B1=B1-1
690 GØ TØ 750
700 LET Z=F(K1)-100
710 IF A(Z)=0 THEN 770
720 IF RND(0)>M(K) THEN 770
730 LET A(Z)=0
740 A1=A1-1
750 IF B1<>0 THEN 770
760 IF A1=0 THEN 850
770 NEXT K1
780 NEXT K
800 S=S+B1
830 T=T+A1
850 NEXT J
860 Q1=10*(1+W)
870 Q2=M1*Q1
880 Q3=10*M1
890 Q4=W*Q3
900 PRINT,1000,I/Q1,(S+T)/Q2,T/Q4,S/Q3
950 NEXT I
960 PRINT
970 PRINT
980 PRINT
990 GØ TØ 120
1000 FMT F9.2,X24,F7.3,X6,F7.3,X6,F7.3
1010 END
```

INSTRUCTIONS FOR DLMNTY

100-110 Dimension F, B and A.

120-230 Input Section.

- 120-130 Input number of passes for interceptors, bomber scale factor (10 times W is number of bombers of type A in each cell) and loop size.
- 140 If 0 was entered for P, then game is over and program goes to END.
- 150-170 Program asks if a new probabilities array is to be set up.
- 180-210 If YES, then a loop is set up, indexed by the number of passes, to input the probabilities of detection and conversion and of kill for each of the two types of bombers.

240-250 Output headers.

310-950 Outer loop, indexed by I/B.

- 320-330 Sum of surviving bombers of each type set to 0.
- 340-850 Iteration loop indexed up to the specified loop size.
  - 350-400 Both bomber arrays B and A are set to 1 (all bombers alive).
  - 410-420 B bomber total/cell set to 10; A bomber total cell set to 10*W.
  - 430-780 PASS loop.
    - 440-460 Interceptors initialized to zero.
    - 470-600 Interceptor assignment loop.
    - 480 Pick a random number and assign the interceptor to the B or A type of bomber based on the ratio of the bombers.

INSTRUCTIONS FOR DLMNTY—Continued

485	If type B all killed, next interceptor.
490	Check if a type B bomber is detected.
510-530	Randomly assign the interceptor to a live type B bomber.
545	If type A all killed, next interceptor.
550	Check if a type A bomber is detected.
570-590	Randomly assign the interceptor to a live type A bomber (100 is added so that the assignment can later check for type A or B.

610-770	Attack loop.
620	If interceptor was not assigned, return interceptor.
630	If assigned to type A, go to 700 (i.e., if assignment number is over 100).
650	If bomber already killed, next interceptor.
660	Check for kill (Monte Carlo).
670	Set bomber = 0.
680	Decrease type B bomber total by 1.
700	Subtract 100 to discover the type A bomber assigned to the interceptor.
710	If all killed, next interceptor.
720	Check for kill.
730	Set bomber = 0.
740	Decrease type A total by 1.
750-760	Check if all bombers of each type killed. If YES, exit to next iteration.
770	End of attack loop.

780	End of pass loop.
800-830	Add surviving bombers to respective totals, by type.

INSTRUCTIONS FOR D'.MNTY—Continued

850 Next iteration.

860-890 Calculation of output scaling factors determined by W and Ml.

900 Print overall I/B, total probability of survival P_s for each type.

950 Next I/B

990 Go to beginning to check for a new game.

1010 END.

VARIABLE LIST FOR DLMNTY

F(*)	Interceptor array
B(*)	Type B bomber array
A(*)	Type A bomber array
P	Number of passes
W	Scale factor for type A bombers
M1	Size of the iteration loop
Y	Flag used to check for new probabilities array
P1	Pass index
D(*)	Probability of Detection and Conversion for type B by pass
C(*)	Probability of Detection and Conversion for type A by pass
L(*)	Probability of kill for type B by pass
M(*)	Probability of kill for type A by pass
I7, I9, I8	Low, high, and step size of I/B ratio
S	Total surviving bombers of type B (summed after each iteration)
T	Total surviving bombers of type A (summed after each iteration)
J, K, K1	Loop indices
B1	Total type B bombers in each cell
A1	Total type A bombers in each cell

VARIABLE LIST FOR DLMNTY—Continued

K2, K3, X	Dummy indices used in interceptor assignment
Q1	Total bombers per cell
Q2	Total bombers used in all iterations
Q3	Total type B bombers used in all iterations
Q4	Total type A bombers used in all iterations

3. Sample Runs of DLMNTY

```
WHAT ARE PASSES, BØMBER SCALE FACTØR, AND LØØP SIZE?!1,2,100
NEW PRØBABILITIES ARRAY?(YES-1,NØ-0)!1
PASS     1 P(DC),P(K) (BØMBER A),P(DC),P(K) (BØMBER B)!.5,.5,.7,.7
I/B RANGE (LØW,HIGH,AND STEP)?!.2,2,.2
         I/B           PRØBABILITY ØF SURVIVAL
                           ALL BØMBERS     TYPE A      TYPE B
     .20                          .940       .951        .918
     .40                          .878       .906        .823
     .60                          .814       .859        .724
     .80                          .764       .809        .674
    1.00                          .728       .783        .619
    1.20                          .667       .732        .538
    1.40                          .625       .684        .508
    1.60                          .611       .691        .449
    1.80                          .557       .633        .405
    2.00                          .525       .605        .366


WHAT ARE PASSES, BØMBER SCALE FACTØR, AND LØØP SIZE?!1,2,100
NEW PRØBABILITIES ARRAY?(YES-1,NØ-0)!1
PASS     1 P(DC),P(K) (BØMBER A),P(DC),P(K) (BØMBER B)!.7,.7,.5,.5
I/B RANGE (LØW,HIGH,AND STEP)?!.2,2,.2
         I/B           PRØBABILITY ØF SURVIVAL
                           ALL BØMBERS     TYPE A      TYPE B
     .20                          .923       .910        .950
     .40                          .848       .817        .910
     .60                          .773       .739        .840
     .80                          .720       .673        .814
    1.00                          .660       .598        .783
    1.20                          .622       .562        .742
    1.40                          .576       .512        .705
    1.60                          .528       .449        .686
    1.80                          .496       .421        .645
    2.00                          .455       .373        .617
```

```
WHAT ARE PASSES, BØMBER SCALE FACTØR, AND LØØP SIZE?!2,2,100
NEW PRØBABILITIES ARRAY?(YES-1,NØ-0)!1
PASS      1 P(DC),P(K) (BØMBER A),P(DC),P(K) (BØMBER B)!.5,.5,.7,.7
PASS      2 P(DC),P(K) (BØMBER A),P(DC),P(K) (BØMBER B)!.3,.3,.5,.5
I/B RANGE (LØW,HIGH,AND STEP)?!.2,2,.2
         I/B                  PRØBABILITY ØF SURVIVAL
                                 ALL BØMBERS       TYPE A         TYPE B
      .20                               .909            .937           .854
      .40                               .814            .867           .707
      .60                               .737            .806           .600
      .80                               .661            .729           .526
     1.00                               .579            .663           .409
     1.20                               .539            .651           .315
     1.40                               .474            .574           .275
     1.60                               .433            .549           .201
     1.80                               .375            .494           .138
     2.00                               .343            .459           .110


WHAT ARE PASSES, BØMBER SCALE FACTØR, AND LØØP SIZE?!0,0,0
```

* The 0,0, 0 input designates exit from DLMNTY.

4. Plot Subroutine—DLMNTP

Following is the program DLMNTP which, when merged with DLMNTY into a program called DLMNTZ, plots the total bomber P_S versus I/B. The P_S for each bomber type is also printed out.

```
240 PRINT
250 PRINT TAB(2);"PS";TAB(40);"PRØBABILITY ØF SURVIVAL"
260 PRINT TAB(17);"0";TAB(27);".1";TAB(37);".2";TAB(47);".3";TAB(57);".4
270 PRINT TAB(67);".5";TAB(77);".6";TAB(87);".7";TAB(097);".8";
280 PRINT TAB(107);".9";TAB(117);"1"
290 PRINT TAB(16);"I----+----I----+----I----+----I----+----I----+";
300 PRINT "----I----+----I----+----I----+----I----+----I----+----I"
302 PRINT TAB(16);"I"
304 PRINT TAB(16);"I"
900 PRINT TAB(16);"I"
910 PRINT,1000,(S+T)/Q2,I/Q1;
915 PRINT" +";
920 PRINT TAB(17+INT(100*(((S+T)/Q2)+.005)));"*"
930 PRINT T/Q4;TAB(11);"(A)   I"
940 PRINT S/Q3;TAB(11);"(B)   I"
1000 FMT F5.3,F10.2
```

SAMPLE RUNS USING DLMNTP
(Merge DLMNTZ, DLMNTY, DLMNTP)

```
MERGE DLMNTZ,DLMNTY,DLMNTP

?XBASIC DLMNTZ
RUN
WHAT ARE PASSES, BOMBER SCALE FACTOR, AND LOOP SIZE?!1.2.100
NEA PROBABILITIES ARRAY?(YES-1,NO-0)?1
PASS     1 P(DC),P(K) (BOMBER A),P(DC),P(K) (BOMBER B)!.5,.5,.7,.7
I/B RANGE (LOW,HIGH,AND STEP)?!.2,2,.2
```




```
WHAT ARE PASSES, BOMBER SCALE FACTOR, AND LOOP SIZE?!
```

SAMPLE RUNS USING DLMNTP—Continued
(Merge DLMNTZ, DLMNTY, DLMNTP)

```
XBASIC DLMNTZ
RUN
WHAT ARE PASSES, BOMBER SCALE FACTOR, AND LOOP SIZE?!1,2,100
NEW PROBABILITIES ARRAY?(YES=1,NO=0)!1
PASS     1 P(DC),P(K) (BOMBER A),P(DC),P(K) (BOMBER B)!.7,.7,.5,.5
I/B RANGE (LOW,HIGH,AND STEP)?!.2,2,.2

 PS
```

```
.924       .20
     .9065(A)
     .958 (B)
.851       .40
     .8315(A)
     .889 (B)
.777       .60
     .741 (A)
     .849 (B)
.723       .80
     .669 (A)
     .832 (B)
.664      1.00
     .609 (A)
     .774 (B)
.617      1.20
     .5475(A)
     .756 (B)
.560      1.40
     .505 (A)
     .67  (B)
.532      1.60
     .4555(A)
     .685 (B)
.453      1.80
     .4115(A)
     .625 (B)
.455      2.00
     .3725(A)
     .62  (B)
```




```
WHAT ARE PASSES, BOMBER SCALE FACTOR, AND LOOP SIZE?!
```

SAMPLE RUNS USING DLMNTP—Continued
(Merge DLMNTZ, DLMNTY, DLMNTP)

```
XBASIC DLMNTZ
RUN
WHAT ARE PASSES, BOMBER SCALE FACTOR, AND LOOP SIZE?12,2,100
LEM PROBABILITIES ARRAY?(YES=1,NO=0)?1
PASS     1 P(DC),P(K) (BOMBER A),P(DC),P(K) (BOMBER B)?.5,.5,.7,.7
PASS     2 P(DC),P(K) (BOMBER A),P(DC),P(K) (BOMBER B)?.3,.3,.5,.5
I/B RANGE (LOW,HIGH,AND STEP)?.2,2,.2

 PS                                PROBABILITY OF SURVIVAL
                 0    .1   .2   .3   .4   .5   .6   .7   .8   .9   1
```



```
.900        .20
      .937 (A)
      .854 (B)
.814        .40
      .8675(A)
      .707 (B)
.737        .60
      .806 (A)
      .6   (B)
.661        .80
      .729 (A)
      .526 (B)
.579       1.00
      .6615(A)
      .409 (B)
.530       1.20
      .6505(A)
      .315 (B)
.474       1.40
      .5735(A)
      .275 (B)
.433       1.60
      .5475(A)
      .201 (B)
.375       1.80
      .4935(A)
      .134 (B)
.343       2.00
      .459 (A)
      .11  (B)

WHAT ARE PASSES, BOMBER SCALE FACTOR, AND LOOP SIZE?0,0,0
 LINE EXIT
```

IV. TWO WAVE BOMBER-INTERCEPTOR MODEL—TOOTH

A. Purpose

The purpose of the TOOTH series of computer models is to fill the gap between simple Monte Carlo models which relate probabilities of I/B ratios and detailed simulation models involving basing, C^3, geography, and operational specifics. The TOOTH models can be used with basic inputs for rapid excursions or for use with a manual war game to exercise the random parts.

B. Basic Concept of TOOTH

The basic concept involves the penetration by four types of bombers—two carrying gravity weapons and two carrying standoff weapons (ASMs). Defense is provided in two waves by two types of interceptors. For the first attack wave, it is assumed that there is a degrading ECM environment and the C^3 only provides a raid count by cell number. The interceptors then attack by committing a certain number per cell. In the second wave, it is assumed standoff weapons have been launched and that a bomber count has been made. The weapons of interceptor type 2 attack the ASMs and bombers as resources allow. A tally is made of surviving bombers at the end of each wave and of the number of surviving ASMs of each type after the second wave.

The program is a stylized war game because no geographical aspects are modeled. However, the operational aspects of the game are accounted for in the inputs to the program. This

manual "homework" is used to determine the number of interceptors that can be brought to bear in the first wave, the kind of weapons they carry, and such details as turn-around. For example, type 1 interceptors can be added to the second wave "pool" of interceptors to represent that fraction of interceptors that could be readied for the second wave attack. Note that the bombers do not release ASMs until after the first wave of interceptors has attacked. This is a rough approximation of the geometric/geographic aspects of air defense.

In this model, we consider ASMs to be attackable targets and allocate interceptors to the bombers and ASMs on a random basis.

C. Program Description

1. Input

The following numbers are requested from the user to run TOOTH:

- Number of bomber cells (Type 1-4), C(1), C(2), C(3), and C(4).
- Number of interceptors/cell and passes for the 1st and 2nd wave (I1, P1, I2, P2), where I1 is the number of interceptors of type 1 committed to each cell on the 1st wave, P1 is the number of passes of type 1 interceptors against the penetrators, I2 is the <u>total number</u> of interceptors of type 2 assigned to the bombers and ASM's for the 2nd wave, and P2 is the number of passes the type 2 interceptors make against the bombers and ASM's.

The following probabilities are requested in this order:

- K(1,I), K(2,I), K(3,I), K(4,I), where I is the index for the number of passes by type 1 interceptors on the 1st wave. Therefore, K(3,2) would be the probability of kill of the type 1 interceptors against type 3 bomber on the 2nd pass.
- L(1,I), L(2,I), L(3,I), L(4,I), where I is the index for the number of passes by the type 2 interceptors on the 2nd wave.
- K3, K4, where K3 is the P_K of the type 2 interceptor against the type 1 ASM and K4 is the P_K of the type 2 interceptor against the type 2 ASM.
- R3, R4, where R3 is the probability of a successful type 1 ASM launch and R4 is the probability of a successful type 2 ASM launch.

The loop size M1 is to be specified and indicates the number of times the game is to be played.

2. Simulation

The game is played M1 times to improve statistics; the user specifies M1 each time. The results are averages of these M1 games.

In each game, the bomber cells are chosen for attack, one at a time. The number of bombers per cell is randomly chosen between 1 and 10. The number of bombers surviving the first wave of attack is accumulated. A bomber is considered killed or not killed by a random choice based on the probability of kill involved. The number of interceptors committed is already known, so the terminal engagement is repeated until all passes have been made or all bombers are killed. Surviving bombers of each type are tallied for the second wave.

For the second wave, it is assumed that a count of the surviving bombers of each type is known. Also, the number of ASMs launched is known. The program allows those penetrators which carried ASMs to continue with the raid for detection and interception. The program assumes ASMs could be attacked from the total pool of type 2 interceptors. The interceptors are uniformly committed to bombers and ASMs for the specified number of passes till the supply is exhausted or all bombers and/or ASMs are killed.

After the specified number of passes on the second wave, the surviving bombers and ASMs are accumulated and the entire game is reinitialized and is repeated until M1 runs have been made.

3. Output

The output totals are averages from the M1 games. Since only cells were specified on input, the program ouputs the average size of the raid in terms of total penetrators. It is assumed that the averages of each bomber type are proportional to the number of cells of each type.

The summary of results is:

-1 number of bombers in attack and of each type (1-4)

-2 number of bombers surviving 1st wave of attack by type (1-4)

-3 number of bombers surviving 2nd wave of attack by type (1-4)

-4 ASMs launched by bomber types 1 and 2

-5 ASMs surviving by types 1 and 2.

4. Coding

- Program List for TOOTH
- Instructions for TOOTH
- Variable List for TOOTH

PROGRAM LIST FOR TOOTH

```
100 DIM Z(10,10),Y(10,10),A(10,10),B(10,10)
1000 PRINT "NUMBER ØF BØMBER CELLS(TYPES 1 TØ 4)?";
1010 INPUT C(1),C(2),C(3),C(4)
1020 PRINT "NUMBER INTERC/CELL AND PASSES, 1ST AND 2ND WAVE?";
1030 INPUT I1,P1,I2,P2
1040 PRINT "NEW PRØBABILITIES ARRAY (YES-1,NØ-2)?";
1050 INPUT Y
1060 IF Y=2 THEN 1300
1070 FØR I=1 TØ P1
1080 PRINT "PASS"I;"1ST WAVE P(K) VS EACH BØMBER TYPE?";
1090 INPUT K(1,I),K(2,I),K(3,I),K(4,I)
1100 NEXT I
1110 FØR I=1 TØ P2
1120 PRINT "PASS"I;"2ND WAVE P(K) VS EACH BØMBER TYPE?";
1130 INPUT L(1,I),L(2,I),L(3,I),L(4,I)
1140 NEXT I
1150 PRINT "P(K) VS ASM-S,1ST TYPE,2ND TYPE?";
1160 INPUT K3,K4
1190 PRINT "ASM LAUNCH RELIBILITY -1ST TYPE,2ND TYPE?";
1200 INPUT R3,R4
1300 PRINT "LØØP SIZE?";
1310 INPUT M1
1500 MAT B=ZER
1600 MAT A=ZER
1800 FØR M=1 TØ M1
2060 MAT Z=ZER
2070 MAT Y=ZER
2100 FØR T=1 TØ 4
2105 IF C(T)=0 THEN 2390
2110 FØR I=1 TØ C(T)
2120 B9=INT(10*RND(0)+1)
2130 Z(T,3)=Z(T,3)+B9
2135 IF I1=0 THEN 2370
2140 I5=I1
2200 FØR P=1 TØ P1
2210 IF B9>=I1 THEN 2300
2220 I5=B9
2300 FØR L=1 TØ I5
2310 IF RND(0)>K(T,P) THEN 2350
2320 B9=B9-1
2330 IF B9=0 THEN 2370
2350 NEXT L
2360 NEXT P
2370 Z(T,1)=Z(T,1)+B9
2380 NEXT I
2390 NEXT T
```

PROGRAM LIST FOR TOOTH—Continued

```
2500 Y(1,3)=INT(R3*Z(1,1))
2510 Y(2,3)=INT(R4*Z(2,1))
2520 Y(1,1)=Y(1,3)
2530 Y(2,1)=Y(2,3)
2600 FØR J=1 TØ 4
2610 Z(J,2)=Z(J,1)
2700 NEXT J
2900 A9=Y(1,1)+Y(2,1)+Z(1,2)+Z(2,2)+Z(3,2)+Z(4,2)
3000 FØR I=1 TØ 12
3010 Q=INT(6*RND(0)+1)
3020 GØTØ 3090,3190,3290,3390,3490,3590 ØN Q
3090 IF Y(1,1)=0 THEN 3010
3100 IF RND(0)>K3 THEN 3700
3110 Y(1,1)=Y(1,1)-1
3120 A9=A9-1
3130 GØTØ 3700
3190 IF Y(2,1)=0 THEN 3010
3200 IF RND(0)>K4 THEN 3700
3210 Y(2,1)=Y(2,1)-1
3220 A9=A9-1
3230 GØTØ 3700
3290 IF Z(1,2)=0 THEN 3010
3300 IF RND(0)>L(1,1) THEN 3700
3310 Z(1,2)=Z(1,2)-1
3320 A9=A9-1
3330 GØTØ 3700
3390 IF Z(2,2)=0 THEN 3010
3400 IF RND(0)>L(2,1) THEN 3700
3410 Z(2,2)=Z(2,2)-1
3420 A9=A9-1
3430 GØTØ 3700
3490 IF Z(3,2)=0 THEN 3010
3500 IF RND(0)>L(3,1) THEN 3700
3510 Z(3,2)=Z(3,2)-1
3520 A9=A9-1
3530 GØTØ 3700
3590 IF Z(4,2)=0 THEN 3010
3600 IF RND(0)>L(4,1) THEN 3700
3610 Z(4,2)=Z(4,2)-1
3620 A9=A9-1
3630 GØTØ 3700
3700 IF A9=0 THEN 3800
3710 NEXT I
3800 FØR T=1 TØ 4
3810 B(T,2)=B(T,2)+Z(T,2)
3820 B(T,3)=B(T,3)+Z(T,3)
3830 B(T,1)=B(T,1)+Z(T,1)
3840 NEXT T
```

```
3850 A(1,3)=A(1,3)+Y(1,3)
3860 A(2,3)=A(2,3)+Y(2,3)
3870 A(1,1)=A(1,1)+Y(1,1)
3880 A(2,1)=A(2,1)+Y(2,1)
3900 NEXT M
4000 B=(B(1,3)+B(2,3)+B(3,3)+B(4,3))/M1
4010 A=(A(1,3)+A(2,3))/M1
5000 PRINT "TØTAL ATTACK SIZE BØMBERS";B;"ASM-S";A
5100 PRINT "ATTACK SIZE-BY BØMBER TYPE"
5110 PRINT B(1,3)/M1;B(2,3)/M1;B(3,3)/M1;B(4,3)/M1
5200 PRINT "BØMBERS SURVIVING FIRST WAVE"
5210 PRINT B(1,1)/M1;B(2,1)/M1;B(3,1)/M1;B(4,1)/M1
5300 PRINT "BØMBERS SURVIVING SECØND WAVE"
5310 PRINT B(1,2)/M1;B(2,2)/M1;B(3,2)/M1;B(4,2)/M1
5400 PRINT "ASM-S LAUNCHED-BY BØMBER TYPE"
5410 PRINT A(1,3)/M1;A(2,3)/M1
5500 PRINT "ASM-S SURVIVING"
5510 PRINT A(1,1)/M1;A(2,1)/M1
7000 PRINT "END ØF GAME? (YES=1)";
7010 INPUT G
7020 IF G=1 THEN 9999
7100 GØTØ 1000
9999 END
```

INSTRUCTIONS FOR TOOTH

1000 - 1310 Input Section.

1000 - 1010 The number of bomber cells for each type of interceptor. NOTE: Types 1 and 2 are ASM carriers.

1020 - 1030 I1 is the number of type 1 interceptors per cell committed in the final wave and P1 is the number of passes made. I2 is the total number of type 2 interceptors in the second wave, since there are no "cells" in the second wave. P2 are the number of passes made by the type 2 interceptor.

1040 - 1200 If a new array is to be set up, then the request is made for the probability of kill of each interceptor against each bomber type for each pass. Also, the program requests the P_k and reliability of launch of each ASM type.

1300 - 1310 Number of iterations.

1500 - 1600 Bomber arrays are initialized to zero.

1800 - 3900 Outer iteration loop.

2060 - 2070 Bomber arrays zeroed for each iteration.

2100 - 2390 Raid Attack Loop (by type).

2105 If no cells of type T, next cell.

2110 - 2380 Cell Attack Loop (by # cells).

2120 Pick number 1 through 10 randomly for numbers of bombers in the cell.

2130 Sum Bombers by type within iteration.

2135 If no interceptors/cell, skip attack.

2200 Pass Loop.

2210 If more interceptors than bombers, set number of interceptors to number of bombers left (2220).

2300 Interceptor Attack Loop.

2310 Check for Kill.

2320 If kill, decrease.

2330 Check for all bombers killed in current cell.

2350 Next Interceptor.

2360 Next Pass.

2370 Sum bombers surviving first wave.

2380 Next Cell.

2390 Next Bomber Type.

2500 - 2510 Total ASM's launched (based on surviving bombers and launch reliability).

2520 - 2530 Initialize ASM's for second wave.

2600 - 2700 Initialize bombers for second wave.

2900 Total targets for second wave.

3000 Interceptor attack loop.

3010 Pick random number 1 to 6.

3020 Branch to subset of instructions depending on targets.

3090 - 3130, 3190 - 3230: Attacks on the two types of ASMs.

3290 - 3330, 3390 - 3430, 3490 - 3530, 3590 - 3630: Attacks on the four types of bombers. In each case, kill check is made, sum is decreased, and exit to 3700.

3700 If all killed, go to summarizing section (3800).

3710 Next interceptor.

3800 - 3840 Total Bombers and those surviving each wave are added to the total for all iterations.

3850 - 3880 Total ASM's launched and those surviving the 2nd wave are totaled for all iterations.

3900 Next Iteration.

4000 - 4010 Bombers and ASM's averaged.

5000 - 5510 Print out section.

7000 - 7100 Continuation request.

9999 END

VARIABLE LIST FOR TOOTH

Z (**) Number of bombers for each iteration

B (**) Running total of bombers

Y (**) Number of ASM's for each iteration

A (**) Running total of ASM's

NOTE: In the above arrays, the first position represents the type of bomber or ASM. The second position designates the situation as follows:

-1 those which survive the first pass

-2 those which survive the second pass

-3 total at the start of the game.

Thus:

Z (3,2) is the number of bombers of type 3 which survive the second pass for that iteration.

A (1,3) is the total number of ASM's of type 1 which have been launched in all iterations up to that point.

The A's and B's are the running totals of the individual Z's and Y's. At the end of all iterations, they are averaged to find the results of the "average" game.

C (*) Number of cells of each type (1-4)

I1 Number of interceptors/cell committed first wave

I2 Total number of interceptors second wave

P1 Number passes first wave interceptors

P2 Number passes second wave interceptors

K(**) Kill probability array for first wave; 1st position is bomber type, 2nd position is pass.

VARIABLE LIST FOR TOOTH—Continued

L (**)	Same as K (**) for second wave
K3	P_k against 1st ASM type
K4	P_k against 2nd ASM type
R3	Reliability of launch of 1st ASM type
R4	Reliability of launch of 2nd ASM type
M1	LOOP size for iteration
T	Bomber type
B9	Number bombers within the current cell
A9	Total targets for second wave

5. Sample Runs of TOOTH

```
NUMBER OF BOMBER CELLS(TYPES 1 TO 4)?? 15,10,15,10
NUMBER INTERC/CELL AND PASSES, 1ST AND 2ND WAVE?? 3,3,300,3
NEW PROBABILITIES ARRAY (YES-1,NO-2)?? 1
PASS 1     1ST WAVE P(K) VS EACH BOMBER TYPE?? .7,.5,.7,.5
PASS 2     1ST WAVE P(K) VS EACH BOMBER TYPE?? .6,.4,.6,.4
PASS 3     1ST WAVE P(K) VS EACH BOMBER TYPE?? .6,.4,.6,0
PASS 1     2ND WAVE P(K) VS EACH BOMBER TYPE?? .6,.4,.6,.4
PASS 2     2ND WAVE P(K) VS EACH BOMBER TYPE?? .5,.3,.5,.3
PASS 3     2ND WAVE P(K) VS EACH BOMBER TYPE?? .5,.3,.5,0
P(K) VS ASM-S,1ST TYPE,2ND TYPE?? .3,.1
ASM LAUNCH RELIBILITY -1ST TYPE,2ND TYPE?? .8,.8
LOOP SIZE?? 100
TOTAL ATTACK SIZE BOMBERS 268.14  ASM-S 33.61
ATTACK SIZE-BY BOMBER TYPE
 81.29     53.79     78.91     54.15
BOMBERS SURVIVING FIRST WAVE
 19.6      23.47     18.08     31.31
BOMBERS SURVIVING SECOND WAVE
 8.00003E-02     2.88     .18     6.22
ASM-S LAUNCHED-BY BOMBER TYPE
 15.27     18.34
ASM-S SURVIVING
 1.27      11.37
END OF GAME? (YES-1)? 0
```

```
NUMBER OF BOMBER CELLS(TYPES 1 TO 4)?? 15,10,15,10
NUMBER INTERC/CELL AND PASSES, 1ST AND 2ND WAVE?? 3,1,100,3
NEW PROBABILITIES ARRAY (YES-1,NO-2)?? 2
LOOP SIZE?? 100
TOTAL ATTACK SIZE BOMBERS 274.92  ASM-S 76.15
ATTACK SIZE-BY BOMBER TYPE
 82.82     55.17     83.44     53.49
BOMBERS SURVIVING FIRST WAVE
 54.37     41.74     54.88     39.9
BOMBERS SURVIVING SECOND WAVE
 44.13     35.07     44.81     33.22
ASM-S LAUNCHED-BY BOMBER TYPE
 43.14     33.01
ASM-S SURVIVING
 37.86     31.11
END OF GAME? (YES-1)? 0
```

```
NUMBER ØF BØMBER CELLS(TYPES 1 TØ 4)?? 15,10,15,10
NUMBER INTERC/CELL AND PASSES, 1ST AND 2ND WAVE?? 3,1,100,1
NEW PRØBABILITIES ARRAY (YES-1,NØ-2)?? 2
LØØP SIZE?? 100
TØTAL ATTACK SIZE BØMBERS 275.36  ASM-S 75.31
ATTACK SIZE-BY BØMBER TYPE
 82.61     54.67     83.56     54.52
BØMBERS SURVIVING FIRST WAVE
 54.19     40.94     54.97     41.05
BØMBERS SURVIVING SECØND WAVE
 44.14     34.27     45.31     34
ASM-S LAUNCHED-BY BØMBER TYPE
 42.95     32.36
ASM-S SURVIVING
 37.82     30.64
END ØF GAME? (YES-1)? 0
NUMBER ØF BØMBER CELLS(TYPES 1 TØ 4)?? 15,10,15,10
NUMBER INTERC/CELL AND PASSES, 1ST AND 2ND WAVE?? 1,3,100,3
NEW PRØBABILITIES ARRAY (YES-1,NØ-2)?? 2
LØØP SIZE?? 100
TØTAL ATTACK SIZE BØMBERS 279.06  ASM-S 78.7
ATTACK SIZE-BY BØMBER TYPE
 82.61     55.91     84.57     55.97
BØMBERS SURVIVING FIRST WAVE
 55.42     43.84     57.48     47.26
BØMBERS SURVIVING SECØND WAVE
 45.66     37.19     47.14     41.13
ASM-S LAUNCHED-BY BØMBER TYPE
 44.01     34.69
ASM-S SURVIVING
 39.04     32.99
END ØF GAME? (YES-1)? 0
NUMBER ØF BØMBER CELLS(TYPES 1 TØ 4)?? 15,10,15,10
NUMBER INTERC/CELL AND PASSES, 1ST AND 2ND WAVE?? 1,1,100,1
NEW PRØBABILITIES ARRAY (YES-1,NØ-2)?? 2
LØØP SIZE?? 100
TØTAL ATTACK SIZE BØMBERS 274.54  ASM-S 96.08
ATTACK SIZE-BY BØMBER TYPE
 82.22     54.18     83.13     55.61
BØMBERS SURVIVING FIRST WAVE
 71.69     49.27     72.67     49.8
BØMBERS SURVIVING SECØND WAVE
 61.26     42.55     62.66     42.84
ASM-S LAUNCHED-BY BØMBER TYPE
 56.99     39.09
ASM-S SURVIVING
 52.05     37.4
END ØF GAME? (YES-1)? 1
```

6. File Input Subroutine (TOOP)

TOOP can be merged with TOOTH to facilitate input of many excursions. The data is put on a file (in whatever system is used), and then the runs are made in sequence without requiring continuous operator intervention. (NOTE: REM statements are not compiled.)

```
10 FILES *
1000 REM "NUMBER ØF BØMBER CELLS(TYPES 1 TØ 4)?"
1010 READ #1, C(1),C(2),C(3),C(4)
1020 REM "NUMBER INTERC/CELL AND PASSES, 1ST AND 2ND WAVE?"
1030 READ #1, I1,P1,I2,P2
1040 REM "NEW PRØBABILITIES ARRAY (YES-1,NØ-2)?"
1050 READ #1, Y
1080 REM "PASS"I;"1ST WAVE P(K) VS EACH BØMBER TYPE?"
1090 READ #1, K(1,I),K(2,I),K(3,I),K(4,I)
1120 REM "PASS"I;"2ND WAVE P(K) VS EACH BØMBER TYPE?"
1130 READ #1, L(1,I),L(2,I),L(3,I),L(4,I)
1150 REM "P(K) VS ASM-S,1ST TYPE,2ND TYPE?"
1160 READ #1, K3,K4
1190 REM "ASM LAUNCH RELIBILITY -1ST TYPE,2ND TYPE?"
1200 READ #1, R3,R4
1300 REM "LØØP SIZE?"
1310 READ #1, M1
7000 REM "END ØF GAME? (YES-1)";
7010 READ #1,G
```

V. EPILOGUE

The three models described in this Note have been in use at ANSER for a number of years in many projects. Their results have proved to be comparable with those of large-scale simulations and expected-value models, even though the large-scale simulations show much more detail and the expected-value models less detail.

The three models are economical. Typical runs cost only a few dollars. Many excursions have been run for comparisons of interceptor missiles and fire control systems. The keys to the results lie in the homework. Essentially, these programs relieve the analyst of the bookkeeping associated with a small war game and give him quick feedback on variations of the input parameters.

There have been many modifications. The models, being simple and short, allow room for imagination and variation.

PRIMARY DISTRIBUTION LIST FOR DDN 73-2

Organization	Number of Copies
AF/RDQL	1
AF/RDQS	1
AF/RDQSD	5
AF/SA	1
AF/SAMI	1
AF/SAS	1
AF/INA	1
ADC/XPA	1
ADC/DOA	1
AFSC/XRT	1
ASD	1
AUL/LSE	1
OSD (DDR&E)	1
ODDPA&E	1
IDA	1
MITRE	1
RAND	2
Washington, D.C.	
Santa Monica, CA	
DDC	2
ANSER	
Library	5
Reserve Stock	10
Master Copy (1)	

UNCLASSIFIED

Security Classification

DOCUMENT CONTROL DATA - R & D

(Security classification of title, body of abstract and indexing annotation must be entered when the overall report is classified)

1. ORIGINATING ACTIVITY (Corporate author)

Analytic Services Inc. (ANSER)
Falls Church, Virginia 22041

2a. REPORT SECURITY CLASSIFICATION

UNCLASSIFIED

2b. GROUP

----

3. REPORT TITLE

QUICK EVALUATION COMPUTER MODELS FOR AIR DEFENSE

4. DESCRIPTIVE NOTES (Type of report and inclusive dates)

Defense Division Note DDN 73-2

5. AUTHOR(S) (First name, middle initial, last name)

Victor V. Gogolak

6. REPORT DATE

May 1973

7a. TOTAL NO. OF PAGES

51

7b. NO. OF REFS

0

8a. CONTRACT OR GRANT NO.

F44620-69-C-0014

b. PROJECT NO.

c.

d.

9a. ORIGINATOR'S REPORT NUMBER(S)

DDN 73-2

9b. OTHER REPORT NO(S) (Any other numbers that may be assigned this report)

10. DISTRIBUTION STATEMENT

Distribution limited to U.S. Government agencies only; test and evaluation; May 1973. Other requests for this document must be referred to Hq. USAF (AF/RDQSD).

11. SUPPLEMENTARY NOTES

12. SPONSORING MILITARY ACTIVITY

Directorate of Operational Requirements and Development Plans, Hq. USAF

13. ABSTRACT

This Note discusses three time-sharing computer programs which are quick tools for air defense engagement analysis. The programs are based on a Monte Carlo method. The model MONTYX evaluates air defense interceptor configurations by determining their effectiveness against a bomber cell in the terminal phase of an intercept. The model DLMNTY expands the capability of MONTYX to a situation where two bomber types are contained within the bomber cell. The results are bomber probabilities of survival for various interceptor to bomber ratios. The model TOOTH involves the penetration by four types of bombers--two carrying gravity weapons and two carrying standoff weapons. The defense is provided by two types of interceptors in two waves. In the first wave, only raid count by cell is provided. The second wave assumes that a bomber count has been made and standoff weapons have been launched. The results indicate the numbers of bombers of each type surviving each wave of defense as well as the number of standoff weapons launched and those which survive the second wave of defense.

Security Classification

14. KEY WORDS	LINK A		LINK B		LINK C	
	ROLE	WT	ROLE	WT	ROLE	WT